pragma solidity 0.4.19;


library SafeMath {
  function mul(uint256 a, uint256 b) internal pure returns (uint256) {
    if (a == 0) {
      return 0;
    }
    uint256 c = a * b;
    assert(c / a == b);
    return c;
  }

  function div(uint256 a, uint256 b) internal pure returns (uint256) {
    // assert(b > 0); // Solidity automatically throws when dividing by 0
    uint256 c = a / b;
    // assert(a == b * c + a % b); // There is no case in which this doesn't hold
    return c;
  }

  function sub(uint256 a, uint256 b) internal pure returns (uint256) {
    assert(b <= a);
    return a - b;
  }

  function add(uint256 a, uint256 b) internal pure returns (uint256) {
    uint256 c = a + b;
    assert(c >= a);
    return c;
  }
}
/**
 * @title Ownable
 * @dev The Ownable contract has an owner address, and provides basic authorization control
 * functions, this simplifies the implementation of "user permissions".
 */
contract Ownable {
  address public owner;


  event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);


  /**
   * @dev The Ownable constructor sets the original `owner` of the contract to the sender
   * account.
   */
  function Ownable() public {
    owner = msg.sender;
  }


  /**
   * @dev Throws if called by any account other than the owner.
   */
  modifier onlyOwner() {
    require(msg.sender == owner);
    _;
  }


  /**
   * @dev Allows the current owner to transfer control of the contract to a newOwner.
   * @param newOwner The address to transfer ownership to.
   */
  function transferOwnership(address newOwner) public onlyOwner {
    require(newOwner != address(0));
    OwnershipTransferred(owner, newOwner);
    owner = newOwner;
  }

}

contract ERC20Basic {
  uint256 public totalSupply;
  function balanceOf(address who) public view returns (uint256);
  function transfer(address to, uint256 value) public returns (bool);
  event Transfer(address indexed from, address indexed to, uint256 value);
}

contract ERC20 is ERC20Basic {
  function allowance(address owner, address spender) public view returns (uint256);
  function transferFrom(address from, address to, uint256 value) public returns (bool);
  function approve(address spender, uint256 value) public returns (bool);
  event Approval(address indexed owner, address indexed spender, uint256 value);
}

contract BasicToken is ERC20Basic {
  using SafeMath for uint256;

  mapping(address => uint256) balances;

  /**
  * @dev transfer token for a specified address
  * @param _to The address to transfer to.
  * @param _value The amount to be transferred.
  */
  function transfer(address _to, uint256 _value) public returns (bool) {
    require(_to != address(0));
    require(_value <= balances[msg.sender]);

    // SafeMath.sub will throw if there is not enough balance.
    balances[msg.sender] = balances[msg.sender].sub(_value);
    balances[_to] = balances[_to].add(_value);
    Transfer(msg.sender, _to, _value);
    return true;
  }

  /**
  * @dev Gets the balance of the specified address.
  * @param _owner The address to query the the balance of.
  * @return An uint256 representing the amount owned by the passed address.
  */
  function balanceOf(address _owner) public view returns (uint256 balance) {
    return balances[_owner];
  }

}

contract StandardToken is ERC20, BasicToken {

  mapping (address => mapping (address => uint256)) internal allowed;


  /**
   * @dev Transfer tokens from one address to another
   * @param _from address The address which you want to send tokens from
   * @param _to address The address which you want to transfer to
   * @param _value uint256 the amount of tokens to be transferred
   */
  function transferFrom(address _from, address _to, uint256 _value) public returns (bool) {
    require(_to != address(0));
    require(_value <= balances[_from]);
    require(_value <= allowed[_from][msg.sender]);

    balances[_from] = balances[_from].sub(_value);
    balances[_to] = balances[_to].add(_value);
    allowed[_from][msg.sender] = allowed[_from][msg.sender].sub(_value);
    Transfer(_from, _to, _value);
    return true;
  }

  /**
   * @dev Approve the passed address to spend the specified amount of tokens on behalf of msg.sender.
   *
   * Beware that changing an allowance with this method brings the risk that someone may use both the old
   * and the new allowance by unfortunate transaction ordering. One possible solution to mitigate this
   * race condition is to first reduce the spender's allowance to 0 and set the desired value afterwards:
   * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
   * @param _spender The address which will spend the funds.
   * @param _value The amount of tokens to be spent.
   */
  function approve(address _spender, uint256 _value) public returns (bool) {
    allowed[msg.sender][_spender] = _value;
    Approval(msg.sender, _spender, _value);
    return true;
  }

  /**
   * @dev Function to check the amount of tokens that an owner allowed to a spender.
   * @param _owner address The address which owns the funds.
   * @param _spender address The address which will spend the funds.
   * @return A uint256 specifying the amount of tokens still available for the spender.
   */
  function allowance(address _owner, address _spender) public view returns (uint256) {
    return allowed[_owner][_spender];
  }

  /**
   * approve should be called when allowed[_spender] == 0. To increment
   * allowed value is better to use this function to avoid 2 calls (and wait until
   * the first transaction is mined)
   * From MonolithDAO Token.sol
   */
  function increaseApproval(address _spender, uint _addedValue) public returns (bool) {
    allowed[msg.sender][_spender] = allowed[msg.sender][_spender].add(_addedValue);
    Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

  function decreaseApproval(address _spender, uint _subtractedValue) public returns (bool) {
    uint oldValue = allowed[msg.sender][_spender];
    if (_subtractedValue > oldValue) {
      allowed[msg.sender][_spender] = 0;
    } else {
      allowed[msg.sender][_spender] = oldValue.sub(_subtractedValue);
    }
    Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

}

contract UCCoin is StandardToken, Ownable {

    string public constant name = "UC Coin";
    string public constant symbol = "UCN";
    uint8 public constant decimals = 8;

    uint256 public INITIAL_TOKEN_SUPPLY = 500000000 * (10 ** uint256(decimals));

    function MAX_UCCOIN_SUPPLY() public view returns (uint256) {
        return totalSupply.div(10 ** uint256(decimals));
    }

    function UCCoin() public {
        totalSupply = INITIAL_TOKEN_SUPPLY;
        balances[msg.sender] = totalSupply;
    }
}

contract UCCoinSales is UCCoin {

    uint256 public weiRaised;

    uint256 public UCCOIN_PER_ETHER = 1540;
    uint256 public MINIMUM_SELLING_UCCOIN = 150;

    bool public shouldStopCoinSelling = true;

    mapping(address => uint256) public contributions;
    mapping(address => bool) public blacklistAddresses;

    event TokenPurchase(address indexed purchaser, uint256 value, uint256 amount);
    event UcCoinPriceChanged(uint256 value, uint256 updated);
    event UcCoinMinimumSellingChanged(uint256 value, uint256 updated);
    event UCCoinSaleIsOn(uint256 updated);
    event UCCoinSaleIsOff(uint256 updated);

    function UCCoinSales() public {

    }
    // users can buy UC Coin
    function() payable external {
        buyUcCoins();
    }
    // users can buy UC Coin
    function buyUcCoins() payable public {
        require(msg.sender != address(0));

        bool didSetUcCoinValue = UCCOIN_PER_ETHER > 0;
        require(!shouldStopCoinSelling && didSetUcCoinValue);
        require(blacklistAddresses[tx.origin] != true);

        uint256 weiAmount = msg.value;

        uint256 tokens = getUcCoinTokenPerEther().mul(msg.value).div(1 ether);

        require(tokens >= getMinimumSellingUcCoinToken());
        require(balances[owner] >= tokens);

        weiRaised = weiRaised.add(weiAmount);

        balances[owner] = balances[owner].sub(tokens);
        balances[msg.sender] = balances[msg.sender].add(tokens);
        // send fund...
        owner.transfer(msg.value);

        contributions[msg.sender] = contributions[msg.sender].add(msg.value);

        TokenPurchase(msg.sender, weiAmount, tokens);
    }

    // convert UC amount per ether -> Token amount per ether
    function getUcCoinTokenPerEther() internal returns (uint256) {
        return UCCOIN_PER_ETHER * (10 ** uint256(decimals));
    }
    // convert minium UC amount to purchase -> minimum Token amount to purchase
    function getMinimumSellingUcCoinToken() internal returns (uint256) {
        return MINIMUM_SELLING_UCCOIN * (10 ** uint256(decimals));
    }

    // the contract owner sends tokens to the target address
    function sendTokens(address target, uint256 tokenAmount) external onlyOwner returns (bool) {
        require(target != address(0));
        require(balances[owner] >= tokenAmount);
        balances[owner] = balances[owner].sub(tokenAmount);
        balances[target] = balances[target].add(tokenAmount);

        Transfer(msg.sender, target, tokenAmount);
    }
    // the contract owner can set the coin value per 1 ether
    function setUCCoinPerEther(uint256 coinAmount) external onlyOwner returns (uint256) {
        require(UCCOIN_PER_ETHER != coinAmount);
        require(coinAmount >= MINIMUM_SELLING_UCCOIN);
        
        UCCOIN_PER_ETHER = coinAmount;
        UcCoinPriceChanged(UCCOIN_PER_ETHER, now);

        return UCCOIN_PER_ETHER;
    }
    // the contract owner can set the minimum coin value to purchase
    function setMinUCCoinSellingValue(uint256 coinAmount) external onlyOwner returns (uint256) {
        MINIMUM_SELLING_UCCOIN = coinAmount;
        UcCoinMinimumSellingChanged(MINIMUM_SELLING_UCCOIN, now);

        return MINIMUM_SELLING_UCCOIN;
    }
    // the contract owner can add a target address in the blacklist. if true, this means the target address should be blocked.
    function addUserIntoBlacklist(address target) external onlyOwner returns (address) {
        return setBlacklist(target, true);
    }
    // the contract owner can delete a target address from the blacklist. if the value is false, this means the target address is not blocked anymore.
    function removeUserFromBlacklist(address target) external onlyOwner returns (address) {
        return setBlacklist(target, false);
    }
    // set up true or false for a target address
    function setBlacklist(address target, bool shouldBlock) internal onlyOwner returns (address) {
        blacklistAddresses[target] = shouldBlock;
        return target;
    }  
    // if true, token sale is not available
    function setStopSelling() external onlyOwner {
        shouldStopCoinSelling = true;
        UCCoinSaleIsOff(now);
    }
    // if false, token sale is available
    function setContinueSelling() external onlyOwner {
        shouldStopCoinSelling = false;
        UCCoinSaleIsOn(now);
    }

    // the contract owner can push all remain UC Coin to the target address.
    function pushAllRemainToken(address target) external onlyOwner {
        uint256 remainAmount = balances[msg.sender];
        balances[msg.sender] = balances[msg.sender].sub(remainAmount);
        balances[target] = balances[target].add(remainAmount);

        Transfer(msg.sender, target, remainAmount);
    }
    // check target Address contribution
    function getBuyerContribution(address target) onlyOwner public returns (uint256 contribute) {
        return contributions[target];
    }
}